





```

org 100h
start:
    mov     ax,12h      ; Poe no modo 640x480
    int     10h

    mov     ax,0A000h   ; Faz ES = 0A000h
    mov     es,ax
    sub     bx,bx      ; BX será o offset!

    mov     dx,03C4h
    mov     ax,0F02h   ; MapMask = 1111b
    out     dx,ax

    mov     dx,03CEh
    mov     ax,8008h   ; BitMask = 10000000b
    out     dx,ax
    mov     ax,0500h   ; Set/Reset = 0101b
    out     dx,ax
    mov     ax,0F01h   ; Enable Set/Reset = 1111b
    out     dx,ax

    mov     al,[byte es:bx] ; carrega os latches da VGA
                                ; note que AL não nos
                                ; interessa!!!
                                ; Isso é necessário pq vamos
                                ; alterar apenas o bit 7. Os
                                ; demais são fornecidos pelos
                                ; latches.

    mov     [byte es:bx],al ; Escreve qualquer coisa...
                                ; AL aqui também não nos
                                ; interessa, já que Set/Reset
                                ; é quem manda os dados para
                                ; os planos de bits.

    sub     ah,ah      ; Espera uma tecla!
    int     16h        ; ... senão não tem graça!!! :)

    mov     ax,3
    int     10h        ; volta p/ modo texto 80x25

    int     20h        ; Fim do prog

end start

```

Explicando a listagem acima: Os quatro planos são habilitados em MapMask... depois habilitamos somente o bit 7 em BitMask, seguido pela habilitação dos quatro bits de "Set/Reset" em "Enable Set/Reset". Uma vez que os quatro planos estão habilitados (por MapMask) e que os quatro bits de "Set/Reset" também estão (via "Enable Set/Reset"), colocamos em "Set/Reset" os quatro bits que queremos que sejam escritos nos planos: 0101b (ou 05h). Pois bem... precisamos apenas carregar os latches e depois escrever na memória do sistema.

Tudo bem, vc diz, mas qual é a grande vantagem?! Ora, ora... temos condições de alterar os quatro planos de bits ao mesmo tempo!! E, melhor ainda, estamos em condição de setar até oito pixels ao mesmo tempo!!!! Experimente trocar a linha:

```

mov     ax,8008h      ; BitMask = 10000000b

```

por:

```
+-----+  
|      mov      ax,0FF08h   ; BitMask = 11111111b      |  
+-----+
```

Você verá oito pixels magenta com uma única escrita na memória do sistema!!

Outra grande vantagem é o ganho de velocidade: Na listagem acima os dados que vão ser colocados nos planos de bits não são fornecidos diretamente pela CPU, mas sim por "Set/Reset" e pelos latches. Assim, a placa VGA não se interessa pelo conteúdo de AL que foi escrito na memória do sistema e não adiciona WAIT STATES, já que esse dado não vai para a memória de vídeo (fica só na memória do sistema!!).

É um grande avanço, né?! Well... próximos avanços nos próximos textos.